Supplementary material for Kirkpatrick and Peischl 2012

Mark Kirkpatrick and Stephan Peischl, May 2012

Introduction

This notebook derives the approximations that are presented in Kirkpatrick and Peischl (2012).

Deriving the basic PDE for survival of the mutant (Eq. (7) in the main text)

We begin with a model that is discrete in time and continuous in space. Write $p_{x,t}$ for the probability that a single copy of the mutant at location x and time t leaves descendants that survive into the indefinite future. Using a standard argument from branching processes, we can write the probability that a mutant leaves no descendants as

$$1 - p_{x,t} = \sum_i f_i(x, t)\left[1 - p^*_{x,t}\right]^i$$

where $f_i(x, t)$ is the probability that an individual at point x at time t leaves i offspring, and $p^*_{x,t}$ is the probability that one of those offspring (randomly chosen) leaves surviving descendants. Let the expected number of offspring produced by an individual at point x at time t be $1 + s_{x,t}$. Assuming a Poisson distribution of offspring number then gives

$$1 - p_{x,t} = \sum_i \frac{1}{i!} \exp[-(1 + s_{x,t})]\,(1 + s_{x,t})^i \left[1 - p^*_{x,t}\right]^i$$

$$1 - p_{x,t} = \exp\left[-(1 + s_{x,t})\,p^*_{x,t}\right] \sum_i \frac{1}{i!} \exp\left[-(1 + s_{x,t})\left(1 - p^*_{x,t}\right)\right]\left[(1 + s_{x,t})\left(1 - p^*_{x,t}\right)\right]^i$$

$$= \exp\left[-(1 + s_{x,t})\,p^*_{x,t}\right]$$

Expanding the right-hand side (r.h.s.) gives

$$1 - p_{x,t} \approx 1 - (1 + s_{x,t})\,p^*_{x,t} + \frac{1}{2}(1 + s_{x,t})^2\,{p^*_{x,t}}^2$$

$$\approx 1 - (1 + s_{x,t})\,p^*_{x,t} + \frac{1}{2}\,{p^*_{x,t}}^2$$

where we assumed that $p^*_{x,t}$ and $s_{x,t}$ are $O(\epsilon)$ and we have dropped terms that are $O\left(\epsilon^3\right)$. The value of $p^*_{x,t}$ can be calculated in terms of the migration kernel $m(.)$:

$$p^*_{x,t} = \int m(y)\,p_{x+y,t+1}\,dx$$

$$\approx p_{x,t+1} + \frac{\sigma^2}{2}\,\partial_x^2\,p_{x,t+1}$$

The second step assumes that the third and higher moments of the dispersal kernal are negligible, as with gaussian dispersal. We now have

$$1 - p_{x,t} \approx 1 - (1 + s_{x,t}) \left(p_{x,t+1} + \frac{\sigma^2}{2} \partial_x^2 p_{x,t+1} \right) + \frac{1}{2} (1 + s_{x,t})^2 p_{x,t+1}^2$$

$$\approx 1 - (1 + s_{x,t}) \left(p_{x,t+1} + \frac{\sigma^2}{2} \partial_x^2 p_{x,t+1} \right) + \frac{1}{2} p_{x,t+1}^2$$

where we have assumed that σ^2 is $O(\epsilon)$ and so ${p_{x,t+1}^*}^2 \approx p_{x,t+1}^2$ to the order of this approximation. Rearranging and again dropping terms that are $O(\epsilon^2)$ then gives

$$p_{x,t+1} - p_{x,t} \approx -s_{x,t} \, p_{x,t+1} - \frac{\sigma^2}{2} \partial_x^2 p_{x,t+1} + \frac{1}{2} p_{x,t+1}^2$$

Since all terms on the r.h.s. are $O(\epsilon^2)$, the change in $p_{x,t}$ with t is small. We are therefore justified in approximating the discrete time process by one in continuous time, giving the PDE

$$\partial_t p_{x,t} = -s_{x,t} \, p_{x,t} + \frac{1}{2} p_{x,t}^2 - \frac{\sigma^2}{2} \partial_x^2 p_{x,t}$$

We interpret $s_{x,t}$ as the intrinsic rate of increase of the mutant at point x at time t.

Fitnesses constant in time and space

When fitnesses are constant in time and space, we have

$$0 = -s \, p + \frac{1}{2} p^2$$

and so

$$p = 2 \, s$$

This is Haldane's classic result.

Fitnesses changing in space but constant in time (Derviation of eq. (9))

Assumptions

With fitnesses constant in time, we have the ODE

$$0 = -s_x \, p_x + \frac{1}{2} p_x^2 - \frac{\sigma^2}{2} \partial_x^2 p_x$$

A general solution for that equation seems impossible to derive. To proceed further we assume that selection intensities are given by:

$$s_x = (1 + s_0) \, \mathrm{Exp}\left(- s_0 \frac{x^2}{2} \right) - 1.$$

In the calcualtions below we approximate fitness by a quadratic function:

$$s_x = s_0\left(1 - \frac{x^2}{2}\right).$$

Space has been scaled such that the mutant has a positive growth rate in the region $\left(-\sqrt{2}\,,\ \sqrt{2}\right)$.

▪ Calculations

The assumption for fitness is

```
sX = s0 (1 - x^2/2);
```

Take the Ansatz for the solution of the ODE to be a gaussian:

```
pX = k Exp[-x^2/(2 v)];
```

This gives the right hand side of the ODE as

```
rhs = -sX * pX + 1/2 pX^2 - σ^2/2 ∂x,x pX;
```

Now expand the expression for p_x as a quadratic around $x = 0$:

```
rhs2 = Collect[Series[rhs, {x, 0, 2}] // Normal, x]
```

$$\frac{k^2}{2} + \frac{k\,\sigma^2}{2\,v} - k\,s_0 + \frac{x^2\left(-2\,k^2\,v - 3\,k\,\sigma^2 + 2\,k\,v\,s_0 + 2\,k\,v^2\,s_0\right)}{4\,v^2}$$

Both terms must vanish, which gives us two equations in our two unknowns:

```
$Assumptions = {s0 > 0};
```

```
kvSolns =
 Solve[
   {Coefficient[rhs2, x, 0] == 0,
    Coefficient[rhs2, x, 2] == 0},
   {k, v}] // Simplify
```

$$\left\{\{k \to 0\},\ \left\{k \to 3\,s_0 + \sqrt{s_0\left(2\,\sigma^2 + s_0\right)}\,,\ v \to \frac{1}{2}\left(1 - \sqrt{1 + \frac{2\,\sigma^2}{s_0}}\right)\right\},\right.$$

$$\left.\left\{k \to 3\,s_0 - \sqrt{s_0\left(2\,\sigma^2 + s_0\right)}\,,\ v \to \frac{1}{2}\left(1 + \sqrt{1 + \frac{2\,\sigma^2}{s_0}}\right)\right\}\right\}$$

Since $v > 0$, the third solution is the one we want. Simplify by assuming $\sigma^2 << 1$:

```
kvSoln = Series[{k, v} /. kvSolns[[3]], {σ, 0, 2}] // Normal // Simplify
```

$$\left\{-\sigma^2 + 2\,s_0,\ 1 + \frac{\sigma^2}{2\,s_0}\right\}$$

Our approximation is therefore

```
pXSoln = pX /. {k → kvSoln[[1]], v → kvSoln[[2]]} // Simplify
```

$$e^{-\frac{x^2 s_0}{\sigma^2+2 s_0}} \left(-\sigma^2 + 2\,s_0\right)$$

In the limit of no dispersal, this result is consistent with Haldane's result, which says that the establishment probability is $2\,s_x$:

```
2 sX - Series[(pXSoln /. σ → 0), {x, 0, 2}] // Normal // Simplify
```

0

Summary of results

Our approximation for the establishment probability is:

$$p_x \;=\; \left(2\,s_0 \,-\, \sigma^2\right) \exp\!\left[-\frac{x^2}{2\,v}\right]$$

where

$$v = 1 \,+\, \frac{\sigma^2}{2\,s_0}$$

Thus the maximum probability of establishment is decreased by an amount σ^2. Swamping results, and the mutant goes extinct, if migration is too strong relative to the mutant's maximum fitness.

The width of p_x is greater than the width of the fitness function (which is scaled to 1).

Fitnesses changing in time and space (Derivation of Eqs. (11) and (12))

Assumptions

Now consider a patch whose width ("variance") is 1 and whose optimum moves in time at velocity c:

$$s_{x,t} \;=\; (1+s_0)\exp\!\left[s_0\,\frac{-(x-c\,t)^2}{2}\right] - 1$$

Space has been scaled such that the width ("variance") of the patch is 1. For concreteness (and without loss of generality) we take $c > 0$.

Again, we approximate fitness by a quadratic function:

$$s_{x,t} \;=\; s_0\left(1 - \frac{(x-c\,t)^2}{2}\right)$$

■ Calculations

The fitness function is

```
sXT = s0 (1 - (x - c t)^2 / 2);
```

The establishment probability is given by the PDE derived above:

$$\partial_t p_{x,t} \;\;=\;\; -s_{x,t}\, p_{x,t} + \tfrac{1}{2}\, p_{x,t}^2 - \tfrac{\sigma^2}{2}\, \partial_x^2\, p_{x,t}$$

Our Ansatz for the solution is a gaussian whose maximum also moves at rate c

$$\texttt{pXT = k * Exp}\left[-\frac{(x - c*t - d)^2}{2\,v}\right];$$

where k, d, and v are constants that we need to solve for.

The right and left sides of the PDE are:

$$\texttt{rhs = -sXT * pXT + }\frac{1}{2}\,\texttt{pXT}^2 - \frac{\sigma^2}{2}\,\partial_{x,x}\texttt{pXT // Simplify}$$

$$\frac{1}{2}\, e^{-\frac{(d+c\,t-x)^2}{v}}\, k \left(k - \frac{1}{v^2} e^{\frac{(d+c\,t-x)^2}{2\,v}} \left(d^2 + 2\,c\,d\,t + c^2\,t^2 - v - 2\,d\,x - 2\,c\,t\,x + x^2\right) \sigma^2 + e^{\frac{(d+c\,t-x)^2}{2\,v}} \left(-2 + c^2\,t^2 - 2\,c\,t\,x + x^2\right) s_0\right)$$

$$\texttt{lhs = }\partial_t\,\texttt{pXT}$$

$$\frac{c\, e^{-\frac{(-d-c\,t+x)^2}{2\,v}}\, k\, (-d - c\,t + x)}{v}$$

A quick check that the units are correct:

$$\texttt{unitSubs = }\{d \to x,\ c \to x,\ v \to x^2,\ k \to 1,\ s_0 \to 1,\ \sigma \to x,\ t \to 1\};$$

```
{lhs, rhs} /. unitSubs // Simplify
```

$$\left\{-\frac{1}{\sqrt{e}},\ \frac{1 - 2\sqrt{e}}{2\,e}\right\}$$

Expand $(r.h.s. - l.h.s.)$ as a quadratic in x around $x = d + c\,t$:

```
diff = Series[rhs - lhs, {x, d + c * t, 2}] // Normal // Simplify
```

$$(-d - c\,t + x)\left(-\frac{c\,k}{v} + d\,k\,s_0\right) + \frac{k\left(k\,v + \sigma^2 + \left(-2 + d^2\right) v\,s_0\right)}{2\,v} - \frac{k\,(d + c\,t - x)^2 \left(2\,k\,v + 3\,\sigma^2 + v\left(d^2 - 2\,(1+v)\right) s_0\right)}{4\,v^2}$$

We now have three equations in three unknowns:

```
$Assumptions =
  {s0 > 0, σ > 0, c > 0};
```

```
kvSolns =
  Solve[
   {Coefficient[diff, x, 0] == 0,
    Coefficient[diff, x, 1] == 0, Coefficient[diff, x, 2] == 0},
   {k,
    v,
    d}];
```

We can identify the correct solution by finding which one gives the right result for $c = \sigma^2 = 0$. It turns out to be the last one:

```
Limit[(k /. kvSolns[[1]] /. d → 0), c → 0] // Simplify
```

0

```
Limit[(k /. kvSolns[[2]] /. d → 0), c → 0] // Simplify
```

$-\infty$

```
Limit[(k /. kvSolns[[3]] /. d → 0), c → 0] // Simplify
```

$3\, s_0 + \sqrt{s_0\,\left(2\,\sigma^2 + s_0\right)}$

```
Limit[(k /. kvSolns[[4]] /. d → 0), c → 0] // Simplify
```

$3\, s_0 - \sqrt{s_0\,\left(2\,\sigma^2 + s_0\right)}$

```
kvSoln = kvSolns[[4]];
```

A first approximation can be obtained by linearizing in $\sigma^2$and c:

```
Series[{k, v, d} /. kvSoln, {σ, 0, 2}, {c, 0, 1}] // Normal // PowerExpand // Simplify
```

$$\left\{-\sigma^2 + 2\, s_0,\ 1 + \frac{\sigma^2}{2\, s_0},\ -\frac{c\,\left(\sigma^2 - 2\, s_0\right)}{2\, s_0^2}\right\}$$

To get the leading order effect of *c* on *k* and *v*, we need to expand to second order in *c*:

```
tmp =
 Series[{k, v, d} /. kvSoln, {c, 0, 2}, {σ, 0, 2}, {s0, 0, 1}] // Normal // Simplify
```

$$\left\{\frac{\left(\sigma^2 - 2\, s_0\right)\,\left(c^2 - 2\, s_0^2\right)}{2\, s_0^2},\ \frac{1}{4}\,\left(4 + \frac{3\, c^2\, \sigma^2}{s_0^3} - \frac{2\, c^2}{s_0^2} + \frac{2\, \sigma^2}{s_0}\right),\ -\frac{c\,\left(\sigma^2 - 2\, s_0\right)}{2\, s_0^2}\right\}$$

That is more easy to read if written as:

$$k = \left(1 - \frac{c^2}{2\, s^2}\right)\left(2\, s - \sigma^2\right), \quad v = 1 + \frac{\sigma^2}{2\, s} - \frac{c^2\left(2\, s - 3\, \sigma^2\right)}{4\, s^3}, \quad d = \frac{c\left(-2\, s + \sigma^2\right)}{2\, s^2}$$

If space is measured on a scale that moves with the patch so that fitness is always maximized at x = 0, our approximation can be written as:

```
p[x_, s_, σ2_, c_] := k[s, σ2, c] Exp[-(x - δ[s, σ2, c])^2 / (2 v[s, σ2, c])]

k[s_, σ2_, c_] := (1 - c^2/(2 s^2)) (2 s - σ2)
```

```
v[s_, σ2_, c_] := 1 + σ2/(2 s) - c^2 (2 s - 3 σ2)/(4 s^3)

δ[s_, σ2_, c_] := c (2 s - σ2)/(2 s^2)

p[x_] := p[x, s, σ^2, c]
```

Here are some examples:

```
Plot[{p[x, 0.1, 0.05, 0], p[x, 0.1, 0.05, 0.05],
   p[x, 0.1, 0.05, 0.1], p[x, 0.1, 0.05, 0.135],
   s[x, 0.1]},
  {x, -3, 3},
 AxesLabel → {x, p}, AxesOrigin → {-3, 0},
 PlotRange → {-0.05, 0.15}, PlotStyle → {{Black}}]
```



■ Summary of results

Our approximation for the establishment probability is:

$$p_{x,t} \;=\; \left(1 - \frac{c^2}{2\,s^2}\right)\left(2\,s - \sigma^2\right)\exp\!\left[-\frac{(x-\delta_x - c\,t)^2}{2\,v}\right],$$

where

$$\delta_x = \frac{c\left(2\,s-\sigma^2\right)}{2\,s^2}$$

and

$$v = 1 + \frac{\sigma^2}{2\,s} - \frac{c^2\left(2\,s-3\,\sigma^2\right)}{4\,s^3}.$$